
Programming
Objects and
Properties

Objectives

• Explain objects, collections, and object models.

• Explore the Object Browser.

• Show how to work with properties and methods.

• Use object variables.

• Explore how to iterate through collections of objects.

What is an Object?

An object is a single entity that you can use in code to work with some part of
an application. It’s a thing that you can control, programmatically. Often an
object represents a part of an application you’re already familiar with, like a
form.

An object maintains its own set of properties and methods, which you’ll work
with in order to control the object.

Key Term

Property An attribute of an object, such as its color or its font name. You

can both set and retrieve most properties of most objects.

Method An action the object “knows” how to take. For example, a user

form, in VBA, knows how to make itself visible using its Show

method.

Object Models

As we’ve said, objects represent different parts of an application. For some
applications, just about every part you can work with, through the user
interface, is represented by an object.

The set of objects within an application is referred to as its object model. An
object model is a logical representation of the way programmable objects are
grouped within an application.

Object models are organized hierarchically. This means there are certain
objects that contain other objects, and those objects may contain other objects
still.

When you’re programming in VBA, you’ll find a number of different object
models: one for VBA itself, another for user forms, and another for the
application you’re working in. In each case, an object model consists of
objects, and a collection of like objects. For example, the Microsoft Excel
object model contains the Application object and a collection of open
Workbook objects.

Key Term

Collection A group of objects all with the same type. A collection may

contain any number of objects, or it may contain none at all.

The Microsoft Forms Object Model

Let’s briefly examine the Microsoft Forms object model which contains both
user form and control objects. That’s right, user forms and controls are objects
you can work with in your code. In fact, that’s what we did in the previous
section when we set the caption property of a user form from code.

The Microsoft Forms object model is a little more complex than Figure 1
shows, but for now this is all you need to worry about. The UserForms
collection contains all open UserForm objects. A UserForm object contains a
Controls collection, which contains all the Control objects on the user form.

Figure 1. The relationship between user forms and controls.

The Object Browser

How do you keep track of all the objects in each object model, and their
properties and methods? One helpful tool is the Object Browser, shown in
Figure 2. The Object Browser lists all of the objects, properties, methods,
constants, and events in an object model.

Figure 2. The Object Browser shows the properties, methods, and events for each
object in an object model.

Opening the Object Browser

To open the Object Browser, open a document in your host application and
follow these steps:

1. Open the Visual Basic Editor.

2. From the View menu choose Object Browser, or press F2.

Search text

Project/Library
list

Classes list

Members
of list

Details
pane

The Parts of the Object Browser

We’ve mentioned the Object Browser before, but we haven’t explained how to
use it. In order to familiarize ourselves with the Object Browser, let’s dissect it
into its various parts.

The Project/Library List

This drop-down list contains all of the object models currently available to
VBA. You can select <All Libraries> to view them all at once, or you can
choose an individual library from the list.

Search Text

To search the Object Browser, enter the text you want to search for and press
ENTER, or click the Search button.

Classes List

The Classes list shows all the available types of objects in the selected type
library. When you select an object from the Classes list, the properties,
methods, and events (or members) for the selected class appear in the
“Members of” list box.

“Members of” List

The “Members of” list displays the methods, properties, events, and constants
that belong to the selected object in the Classes list.

Details Pane

The Details pane at the bottom of the Object Browser provides additional
information on any selected class, property, method, event, or constant.

Figuring Out the Object Hierarchy

Although the Object Browser is a great tool for finding out about an object’s
properties and methods, it’s not much help for understanding the overall
structure of an application’s object model. All of the objects in the object
model appear together in the Classes list, and it’s difficult to see quickly how
they relate to one another in the hierarchy.

You can view the complete object model diagrams in online Help. Follow
these steps to find the object model diagrams for Microsoft Excel:

1. From the Help menu, choose Contents and Index.

2. Click the Contents tab.

3. Open the entry for the Microsoft Excel Visual Basic reference, and double-
click the Shortcut entry to open the Microsoft Excel Visual Basic help
file.

4. Open the Microsoft Excel Visual Basic Reference entry, and choose
Microsoft Excel Objects.

The steps are similar for finding the object model diagrams for other Office
applications.

Working with Objects in VBA

Referring to Objects in Collections

As you have seen, many objects in the Microsoft Excel object model belong to
collections of objects. The Workbooks collection contains all the open
workbooks, and the Worksheets collection contains all the worksheets in the
workbook.

To refer to an object in code, you specify its position in the object model. In
other words, you specify an “address” for an object within the object model so
VBA knows exactly which one you want.

Referring to Objects by Name

You can refer to an object in a collection by its name. For example, suppose
you want to refer to a particular workbook object already open in Microsoft
Excel. You can refer to the Workbook object in the Workbooks collection by
its name:

Application.Workbooks("vba.xls")

This code tells VBA that you want to work with the workbook vba.xls. It
refers to the Workbooks collection of the Application object, then to an
individual workbook within the collection.

Actually, you don’t need to include the Application object in this code because
it contains all of the other objects and collections in Microsoft Excel. We
included it just to show that Workbooks is a collection of the Application
object.

Referring to Objects by Position

Another way to refer to an object in a collection is by its position within the
collection, which is identified by an index number. The first open workbook in
the Workbooks collection will have an index of 1, the second, an index of 2,
and so on. To refer to the first workbook in the collection, specify the index
value:

Workbooks(1)

The problem with using this technique is that if there is more than one
workbook open, there’s no way to know which one you’ll get back. If you
need just any old workbook, it doesn’t matter, but if you need a specific one,
you should probably refer to it by name.

What If You Don’t Know the Object’s Name?

When you’re writing code, you may encounter a situation where you need to
refer to an object, but you won’t know which object until the code is running.
In that case, you still refer to the object in the collection by its name, but you
use a string variable to represent the name of the object.

' Here, strWorkbook is a string variable

' containing the name of the workbook.

Workbooks(strWorkbook)

NOTE Using the bang (!) operator syntax is another way to refer to an
object you may encounter. The bang operator separates the
collection from the name of the object. You can use the bang
syntax to refer to an object in a collection, but there’s no need to,
since the parentheses syntax we discussed previously works
perfectly well. What’s more, the parentheses syntax
accommodates a variable, while the bang syntax does not.

The following table summarizes the different ways to refer to an object in a
collection:

Syntax Notes

Collection("ObjectName") Preferred syntax if you know the name of the
object, or if the object name is stored in a variable.

Collection(Index) Refer to an object by its position in the collection.

Collection!ObjectName Acceptable, but less flexible than parentheses
syntax.

Referring to Collections Within Objects

As we’ve mentioned, collections can contain objects, and objects can contain
other collections. We’ve shown how to refer to an object within a collection,
but that’s only the first step. To get to a specific object in the object hierarchy,
you need to refer to both the objects and the collections.

To refer to a collection contained within an object, you use the dot (“.”)
operator. For example, to refer to the Worksheets collection of a Workbook
object, you would use the following syntax:

Workbooks("vba.xls").Worksheets

Note that the dot operator separates the Workbook object specified by
Workbooks(“vba.xls”) from the Worksheets collection.

Of course, you can then refer to a specific object within the Worksheets
collection, and so on through the object hierarchy:

Workbooks("vba.xls").Worksheets(1)

Sometimes you may see a UserForm object and a Control object separated
only by the dot operator, as in the following example:

UserForm1.CommandButton1

What’s going on here? This is actually a special shortcut you can use to refer
to controls on user forms.

In this case, the command button is a Control object that belongs to the
Controls collection of the user form. The Controls collection is the default and,
in fact, only, collection of a user form. That is, if you don’t explicitly state a
collection, VBA assumes that the object belongs to the Controls collection.
You can also explicitly state an object’s membership of the Controls collection
with the following:

UserForm1.Controls("CommandButton1")

Setting Properties on Objects

Forms and Controls

In general, you need to do two things to property values—set them and retrieve
them. The general syntax for setting property values is:

object.property = value

In this example:

• Object is the name of a form or control. (We’ll discuss other types of
objects later in the course).

• Property is the name of a property.

• Value is the new value for the property and can be a literal value, a
variable, or an expression.

You saw an example of this in previous section:

Me.Caption = txtInput.Text

This statement sets the Caption property of the form to the value of the Text
property of the txtInput control.

When you retrieve a property’s value, you usually want to do something with
the value you get back. Depending on what your code is doing, you have a
variety of choices, including:

• Storing the value in a variable so you can use it later in your procedure.

Dim strCaption As String

strCaption = UserForm1.Caption

• Printing the value to the Debug window.

Debug.Print UserForm1.Caption

• Using the value in another expression or function.

MsgBox "The current caption is " & _

UserForm1.Caption

' or

Call HandleButton(cmdOK.Enabled)

TIP: Remember when VBA sees object.property, in most cases, it interprets this as

a value, meaning you can use this syntax wherever you would normally use a

literal value like a string or a number.

Microsoft Excel Objects

You use the same strategies to set and return properties values for Microsoft
Excel objects that you use for properties of forms and controls. This code
prints the path and filename of the workbook to the Immediate window:

Debug.Print Workbooks("VBA.xls").FullName

To work with an object that’s “deeper” into the object model, you have to
specify the objects above it. For example, to return the Visible property of a
Worksheet object, you refer to the workbook that contains the worksheet, then
to the worksheet itself:

Debug.Print Workbooks("VBA.xls").Worksheets(1).Visible

Because we referred to the worksheet by its index number this code will return
the Visible property for the first worksheet in the workbook.

Shorten Your Code with Default Properties

Most objects have a “most likely to be set or retrieved” property called the
default property.

Some objects have as their default property a property that returns an object or
a collection. For example, you use the Controls property to return a reference
to the Controls collection, which is the default collection of the UserForm
object. In other words, the default property and the default collection for the
UserForm object are the same thing.  You can refer to the value of a control on
a user form in one of the following ways:

Debug.Print UserForm1.Controls("txtInput").Value

' or

Debug.Print UserForm1.txtInput.Value

For controls, you’ll most often use the default property, such as the Value
property of the TextBox control. You can get the value of the Value property
without having to actually including the Value property:

Debug.Print txtInput.Value

' or

Debug.Print txtInput

How can you tell which property is an object’s default property? Take a look
at the MSForms library in the Object Browser. The Controls property of the
UserForm object has a small blue marble above it denoting the default
property.

Taking Action: Methods

In addition to properties, the objects in an object model also have methods
which allow an object to perform an action.

Using a method of an object is similar to calling any other procedure, except
you must refer to the specific object. Like procedures, methods can also take
arguments.

Some methods behave like function procedures and return a value or, in some
cases, an object. You can use the value they return elsewhere in your code, or
you can ignore it. Other methods behave like Sub procedures; they simply
perform an operation and don’t return a value.

For example, the Save method of a Workbook object doesn’t return a value, so
you don’t include parentheses with it:

Workbooks("VBA.xls").Close xlSaveChanges

The CheckSpelling method of the Microsoft Excel Application object, on the
other hand, does return a value:

Dim blnSpell As Boolean

blnSpell = Application.CheckSpelling("alot")

In this case, the CheckSpelling method returns False, since the word is spelled
incorrectly.

Using Object Variables

Introducing Object Variables

When manipulating objects, it is often more convenient to use a variable to
represent the object than to refer to it repeatedly throughout the VBA
hierarchical syntax. In fact, for repeated references to the same object, using a
variable results in faster execution of the code.

When you create an object variable, the VBA-created variable only refers to,
or points to, the real object in memory, not to the real object itself. This is
different from other variable types, where the variable itself does actually
contain the value assigned to it.

You can create an object variable to refer to most objects. VBA supports
object variable types for each of the objects allowing you to write code like
this:

Dim frm As UserForm

Dim ctl As Control

Dim wbk As Workbook

Dim wks As Worksheet

In the most general case, you can write:

Dim obj As Object

But the more specific you can be about your object variable declarations, the
more VBA can do for you internally.

In the previous code example, the variables do not hold any data, nor do they
refer to any real objects at this point. To make an object variable actually refer
to a real object, you must use the Set keyword. In every case, you use Set to
point the variable at a real object. For example, using the variables in the
previous example, you might see code like this:

Set frm = UserForms("frmChangeProps")

Set ctl = Me.cmdOK

Set wks = Workbooks("VBA.xls").Worksheets(1)

See
frmReferToObjects

In each case, you’ve made the object variable refer to an actual object. Without
this step, the object variables are just placeholders – they don’t actually refer to
anything.

Using Object Variables

Suppose you want to change a number of properties of a specific control at run
time. The sample form, frmChangeProps, includes code to change properties
of one of the text boxes on the user form. The code uses direct, explicit
references to the control each time:

Me.txtInput.BackColor = RGB(0, 255, 0)

Me.txtInput.ForeColor = RGB(255, 0, 0)

Me.txtInput.FontName = "Arial"

Me.txtInput.FontSize = 14

Me.txtInput.FontBold = True

A better way would be to use an object variable to refer to the control that’s
being altered:

Dim txt As Control

Set txt = Me.txtInput

txt.BackColor = RGB(0, 255, 0)

txt.ForeColor = RGB(255, 0, 0)

txt.FontName = "Arial"

txt.FontSize = 14

txt.FontBold = True

Note the use of the Set keyword to make the variable txt refer to the real text
box before the code attempts to use txt. If you want to see what happens
without the Set statement, comment out that line of code and attempt to run the
form example again. You’ll get an error 91: “Object variable or With block
variable not set.”

Special VBA Constructs for Objects

Introducing For Each...Next

The For Each...Next looping control structure allows you to loop through all
the elements of any collection, without having to know how many items are in
the collection. The general syntax is:

For Each element In group

[statements]

[Exit For]

[statements]

Next [element]

The following points explain each part of the construct:

• element: an object variable of the type of object contained in the
collection.

• group: a collection of objects, such as UserForms, Controls,
Worksheets, etc.

• statements: zero or more statements to execute for each element of the
collection.

• Exit For: just as with For...Next loops, you can exit a
For...Each...Next loop early. The code will continue execution with the
line after the Next [element] statement.

• Next [element]: concludes the loop. The [element] is optional, but
makes the code easier to read, and is highly recommended.

A Simple Example

For example, you could use the following code to loop through all the open
forms and change all their background colors to a specific color:

See frmForEach

Private Sub cmdForEach_Click()

Dim ctl As Control

For Each ctl In UserForms.Controls

ctl.BackColor = RGB(0, 255, 0)

Next ctl

End Sub

Using With...End With

The With ... End With statement allows you to set multiple properties or call
multiple methods on the same object in one construct. For example, suppose
you want to refer to many properties or methods for one object. You can use
the With ...End With statement to set multiple properties, or execute methods,
by typing the object name only once.  The general syntax looks like this:

With object

[statements]

End With

The object reference can be any object, and [statements] refers to any
properties or methods of the referenced object. In those statements, you don’t
repeat the name of the object, or the reference to it; you just include the “.”
separator.

Previously, the code in frmChangeProps changed multiple properties of a
control. Using the With...End With construct, you could rewrite the code like
this:

Dim txt As Control

Set txt = Me.txtTest

With txt

.BackColor = vbBlack

.ForeColor = vbCyan

.FontName = "Arial"

.FontSize = 16

.FontBold = True

End With

See
frmChangeProps

Summary

• Objects represent different parts of a component or application and can
be manipulated with code.

• Objects have properties and methods you can use to determine their
characteristics and make them perform certain actions.

• You can use the Object Browser to investigate properties and methods
of objects.

• To work with an object, you must refer to it by its position in the
object model.

• You can refer to an object either by its name or by its index number in
the collection to which it belongs.

• When creating object variables, assign them the most specific type
possible.

• Use the Set keyword to assign an object variable to an object.

• Once you use the Set keyword, the object variable acts as a reference
to the real object. Use the variable any place you might otherwise use
the object reference.

• Use the For Each…Next construct to loop through all of the objects in
a collection, without having to know the number of objects.

• Use With...End With to set multiple properties or call multiple
methods for a single object, without having to retype its name or make
repeated references to it.


